

Lab1 系统引导实验报告

vigorweijia@foxmail.com

实验框架解读

实验环境

框架代码结构如下：

```
weijianan@debian:~/lab1-15122XXXXXX/lab$ tree
.
├── app.s
├── Makefile
├── boot.c
├── boot.h
├── Makefile
├── start.s
├── Makefile
└── genboot.pl
3 directories, 8 files
```

实验要求实现一个简单的引导程序，并在显示器上打印相应内容。

按照 OS 网站上的教程，系统启动时，工作在实模式下，BIOS 会首先取得控制权，自检各个硬件是否正常工作，然后按 CMOS RAM 中设置的启动设备查找顺序来寻找可启动设备。工作在实模式的 BIOS 程序将主引导扇区（Master Boot Record，也就是磁盘 0 号扇区）加载至内存 0x7c00 处（被加载的程序一般称为 Bootloader），紧接着执行一条跳转指令，将 CS 设置为 0x0000，IP 设置为 0x7c00，运行被装入的 Bootloader。

```
1 os.img:
2   cd bootloader; make bootloader.bin
3   cd app; make app.bin
4   cat bootloader/bootloader.bin app/app.bin > os.img
5
6 clean:
7   cd bootloader; make clean
8   cd app; make clean
9   rm -f os.img
10
11 play:
12   qemu-system-i386 os.img
13
14 debug:
15   qemu-system-i386 -s -S os.img
```

需要执行这些指令，就必须有相应的硬件，本次实验使用的是硬件模拟软件 qemu。为了让 qemu 能够识别 x86 汇编代码，需要进行一系列的参数设定，如图，在根目录下的 Makefile 中 qemu 模拟的是 i386 架构，保证我们的实验能够顺利进行。

Makefile 文件解析

根目录下的 make 文件执行了 bootloader 中的 make 和 app 中的 make，然后强制把得到的 bootloader.bin 和 app.bin 拼接成 os.img，也就是我们实现的简单的系统内存镜像文件。由此，继续解读 bootloader 和 app 文件夹下的 Makefile 文件。

接着是链接生成 `bootloader.elf` 文件。`-m elf_i386` 的意思是生成 i386 架构下的 `elf` 可执行文件。`-e start` 会解析“start”这个符号，并把它作为程序的入口地址。`-Ttext 0x7c00` 意思是把文件重定向到 `0x7c00` 处，这个可以通过 `objdump` 查看，如下：

接着是使用 `objcopy` 命令把 `bootloader.elf` 的内容拷贝到 `bootloader.bin` 中。由于无法对 `bootloader.bin` 使用 `objdump` 命令，使用 `hexedit` 工具查看它的二进制代码：

目下搖回垂穴位信自和筭且信自：，，，音用目口搖回，，，共

然后查看 app 的 Makefile 文件。

如同 bootloader 的 Makefile 文件, 会把 start 作为程序的入口, 并把程序重定向到 0x8c00 这个位置, 最后把 text 段拷贝到 app.bin 文件中。

框架代码解读

程序的入口是 start.s 中的 start, 因此解读从 start.s 开始。

```
1 /* Real Mode Hello World */
2 #.code16
3 #
4 #.global start
5 #start:
6 #     movw %cs, %ax
7 #     movw %ax, %ds
8 #     movw %ax, %es
9 #     movw %ax, %ss
10 #     movw $0x7d00, %ax
11 #     movw %ax, %sp # setting stack pointer to 0x7d00
12 #     pushw $13 # pushing the size to print into stack
13 #     pushw $message # pushing the address of message into stack
14 #     callw displayStr # calling the display function
15 #loop:
16 #     jmp loop
17 #
18 #message:
19 #     .string "Hello, World!\n\n"
20 #
21 #displayStr:
22 #     pushw %bp
23 #     movw 4(%esp), %ax
24 #     movw %ax, %bp
25 #     movw 6(%esp), %cx #length of the string
26 #     movw $0x1301, %ax #ah = 0x13 print the string
27 #     movw $0x000c, %bx #bh = 0x00 black background bl = 0x0c red character
28 #     movw $0x0000, %dx #print ar line 0 col 0
29 #     int $0x10
30 #     popw %bp
31 #     ret
```

start 代码最开始处是被注释掉的实模式下输出 hello world。可以发现, 在设定栈帧位置 (0x7d00) 后, 就开始压入参数, 执行打印字符串的函数 displayStr。int \$0x10 实现的是 BIOS 中断, 要求 %ah = 0x13; %ax = 0x01 表示字符串只包含字符码, 显示之后不更新光标位置, 属性值在 BL 中。

接着是在保护模式下打印 Hello world。

```
33 /* Protected Mode Hello World */
34 #.code16
35 #
36 #.global start
37 #start:
38 #     movw %cs, %ax
39 #     movw %ax, %ds
40 #     movw %ax, %es
41 #     movw %ax, %ss
42 #     cli # clear interruption
43 #     inb $0x92, %al # Fast setup A20 Line with port 0x92, necessary or not?
44 #     orb $0x02, %al
45 #     outb %al, $0x92
46 #     data32 addr32 lgdt gdtDesc # loading gdt, data32, addr32
47 #     movl %cr0, %eax
48 #     orb $0x01, %al
49 #     movl %eax, %cr0 # setting cr0
50 #     data32 jmp $0x08, $start32 # reload code segment selector and jmp to start32, data32
51 #
52 #code32
53 #start32:
54 #     movw $0x10, %ax # setting data segment selector
55 #     movw %ax, %ds
56 #     movw %ax, %es
57 #     movw %ax, %fs
58 #     movw %ax, %ss
59 #     movw $0x10, %ax # setting graphics data segment selector
60 #     movw %ax, %gs
61 #
62 #     movl $0x0000, %eax # setting esp
63 #     movl %eax, %esp
64 #     jmp to bootMain in boot.c
65 #     calll bootMain
66 #     pushl $1
67 #     pushl $message
68 #     calll displayStr
69 #loop32:
70 #     jmp loop32
71 #
72 #message:
73 #     .string "Hello, World!\n\n"
74 #
75 #displayStr:
76 #     movl 4(%esp), %ebx
77 #     movl 8(%esp), %ecx
78 #     movl $((80*540)*2), %edi #print hello world at line 5
79 #     movb $0x0c, %ah #black background and red characters
80 #nextChar:
81 #     movb (%ebx), %al
82 #     movw %ax, %gs:(%edi) # write graphics memory
83 #     addl $2, %edi # %edi is the address of the character
84 #     incl %ebx
85 #     loopnz nextChar # loopnz decrease ecx by 1
86 #     ret
87 #
88 #.p2align 2
89 #gdt: # 8 bytes for each table entry, at least 1 entry
90 #     .word 0,0 # empty entry
91 #     .byte 0,0,0,0
92 #     .word 0xffff,0 # code segment entry
93 #     .byte 0,0x9a,0xcf,0
94 #     .word 0xffff,0 # data segment entry
95 #     .byte 0,0x92,0xcf,0
96 #     .word 0xffff,0x8000 # graphics segment entry
97 #     .byte 0x0b,0x92,0xcf,0
98 #
99 #gdtDesc: # 6 bytes in total
100 #     .word (gdtDesc - gdt - 1) # size of the table, 2 bytes, 65536-1 bytes, 8192 entries
101 #     .long gdt # offset, i.e. linear address of the table itself
102 #
```

首先使用 cli 指令关中断, 启动 A20 总线, 查阅资料得知, 8086 模式下, A20 打开的情

况下，访问超过 1MB 内存，就真实的访问。然后加载 GDTR，设置 CRO 的 PE 位为 1，表示打开分段。通过长跳转设置 CS 进入保护模式，初始化 DS, ES, FS, GS, SS 这几个段寄存器。这里设置了三个 GDT 表项，其中代码段与数据段的基址都为 0x0，视频段的基址为 0xb8000。

这之后，把参数压栈，转而执行 displayStr 函数输出字符串。%ebx 保存了字符串的首地址；%ecx 保存了需要打印的长度；%edi 保存了打印的位置，一行有 80 个字符，将在第 5 行第 0 列打印 hello world，查阅资料得知一个字符占用两个字节的显存，其原因是还需要一个字节保存一些属性，即背景色和字符颜色；%ah 保存背景色和字符颜色。接着就是在 nextChar 这个循环中输出字符串。ret 之后进入 loop32 这个死循环，否则会继续执行下面的代码，导致出错。

神奇的是，字符串的内容是直接保存在代码中的，readelf 找不到数据区中相应的内容，这可以直接从 objdump 或者 hexedit 工具读出。

```
Section Headers:
[Nr] Name              Type              Addr      Off      Size    ES Flg Lk Inf Al
[ 0]                     NULL              00000000  000000  000000  00  0  0  0
[ 1] .text                PROGBITS          00007c00  000c00  000145  00  AX  0  0  4
[ 2] .eh_frame            PROGBITS          00007f48  000c48  000070  00  A   0  0  4
[ 3] .comment              PROGBITS          00000000  000db8  00002d  01  MS  0  0  1
[ 4] .symtab               SYMTAB            00000000  000de8  000140  10   5  13  4
[ 5] .strtab               STRTAB            00000000  000f28  00007a  00   0  0  1
[ 6] .shstrtab             STRTAB            00000000  000fa2  000034  00   0  0  1

Program Headers:
Type      Offset    VirtAddr    PhysAddr   FileSiz MemSiz  Flg Align
LOAD      0x000000 0x00007000  0x00007000 0x000db8 R E 0x1000
GNU_STACK 0x000000 0x00000000  0x00000000 0x000000 RWE 0x10

00007c54 <message>:
7c54: 48          dec    %eax
7c55: 65 6c      gs insb (%dx), %es:(%edi)
7c57: 6c         insb   (%dx), %es:(%edi)
7c58: 6f         insb   (%esi), (%edx)
7c59: 2c 20      sub    $0x20, %eax
7c5b: 57         push   %edi
7c5c: 6f         %edx: (%esi), (%edx)
7c5d: 72 6c      jbe    <readSec+0x13>
7c5f: 64 21 0a   and    %ecx, %fs:(%edx)
...
```



即符号表、只读数据区找不到存储的字符串。
直接执行这段代码，就可以在保护模式下输出 hello world。
由于中断关闭，无法通过陷入磁盘中断调用 BIOS 进行磁盘读取，本次实验通过 readSec(void *dst, int offset)这一接口读写（in, out 指令）磁盘的相应端口（Port）来实现磁盘特定扇区的读取。

```
1 #include "boot.h"
2
3 #define SECTSIZE 512
4
5 void bootMain(void) {
6     void (*elf)(void);
7     elf = (void*)(void)0x8c00;
8     readSec((void*)elf, 1); // loading sector 1 to 0x8c00
9     elf(); // jumping to the loaded program
10 }
11
12 void waitDisk(void) { // waiting for disk
13     while((inByte(0x1f7) & 0xc0) != 0x40);
14 }
15
16 void readSec(void *dst, int offset) { // reading a sector of disk
17     int i;
18     waitDisk();
19     outByte(0x1f2, 1);
20     outByte(0x1f3, offset);
21     outByte(0x1f4, offset >> 8);
22     outByte(0x1f5, offset >> 16);
23     outByte(0x1f6, (offset >> 24) | 0xe0);
24     outByte(0x1f7, 0x20);
25
26     waitDisk();
27     for (i = 0; i < SECTSIZE / 4; i++) {
28         ((int *)dst)[i] = inLong(0x1f0);
29     }
30 }
```

readSec 里面有许多看起来奇怪的向指定端口写入的指令。
查阅资料得知，这是 LBA28 方式读取磁盘，使用 28 个比特位来表示逻辑区号（每 512 字节为 1 个扇区），一共有 8 个端口，编号为 0x1f0~0x1f7。读入的长度放入 1f2（读入的扇区个数），1f3~1f6 保存的是逻辑区号，因此 offset 可以理解为需要读入的起始的磁盘扇区号。0x1f0 是硬盘接口的数据端口，并且是 16 位的，0x1f1 是端口的错误寄存器，里面保留

着“硬盘驱动器”最后一次执行工作的时候状态。（错误原因）

0x1F7是命令端口，又是状态端口。

发送0x20，硬盘里面的“管理员”就开始做一些准备工作了，并且把0x1F7里面的第7个位置变成1代表自己很忙。

0010 0000 =我要拿东西

1010 0000 =我开始准备了

0010 1000 =我准备好了

waitDisk 等待磁盘空闲，此时就通过 inLong 函数读入磁盘内容并写到 dst 里，也就是内存起始地址为 0x8c00 的位置，最后 app.s 的内容从磁盘被读入内存。

在 app 的 Makefile 文件中，用户程序会被写入地址为 0x8c00 的位置，因此 elf() 会跳转到用户程序中。

由此，框架分析完毕，我们需要在 start.s 中跳转到 bootMain，并在 app.s 中写入输出 hello world 的代码。

代码编写

当不需要返回时，可以直接在 start.s 中写入 jmp bootMain，为了保持原结构，我写的是 call bootMain。

```
.code32
start32:
    movw $0x10, %ax # setting data segment selector
    movw %ax, %ds
    movw %ax, %es
    movw %ax, %fs
    movw $0x18, %ax # setting graphics data segment selector
    movw %ax, %gs
    movl $0x0000, %eax # setting esp
    movl %eax, %esp
    # jmp to bootMain in boot.c
    call bootMain
    pushl $1
    pushl $message
    call displayStr
```


在 app.s 中，可以仿照 start.s 中保护模式的代码打印 hello world。不过为了维护栈帧的完整，并且在执行完后能够返回 bootMain 进而返回 start 中，需要创建一个新的栈帧，结构为: push %ebp; mov %esp,%ebp; ... ; leave; ret;

由于已经知道在输出时各个寄存器的作用，可以把代码改写得复杂一点，即在两行输出，并且输出不同的颜色。

```
1 .code32
2
3 .global start
4 start:
5     # TODO
6     push %ebp           #Create new stack frame
7     mov %esp, %ebp
8     pushl $16           #length of the string
9     pushl $message      #address of the string
10    calll displayStr
11    leave
12    ret                 #Recover stack
13 message:
14     .string "hello, NJUworld!\n\0"
15 displayStr:
16     movl 4(%esp), %ebx
17     movl 8(%esp), %ecx
18     subl $0xa, %ecx     #print 6 characters
19     movl $0xa, %edx
20     movl $((80*10+30)*2), %edi #print at line 10, col 30
21     movb $0x0a, %ah     #color setting
22 nextChar1:
23     movb (%ebx), %al
24     movw %ax, %gs:(%edi)
25     addl $2, %edi
26     incl %ebx
27     loopnz nextChar1
28     movl %edx, %ecx     #print 10 characters
29     movl $((80*11+30)*2), %edi
30     movb $0x0e, %ah     #set new color
31 nextChar2:
32     movb (%ebx), %al
33     movw %ax, %gs:(%edi)
34     addl $2, %edi
35     incl %ebx
36     loopnz nextChar2
37     ret
```

每次执行 loopnz 时，%cx = %cx-1，如果 %cx != 0 并且 zf = 0，会跳转到后面的标签，执行循环。

执行结果如下：



红色的 hello world 是保护模式下的输出，绿色和黄色的 hello world 是从磁盘加载的程序的输出。

心得体会

刚开始进行实验的时候比较茫然，甚至不理解要干什么，从哪里入手。仔细阅读了助教的 ppt、网站上的实验教程以及代码框架后，对整个程序的执行流程有了一个比较清晰的认识，然后才写出了代码，这也验证了助教说的读代码两小时，写代码 5 分钟。

但是，对于本次实验，还是有疑问，首先是显存的问题，每个字符占用两字节，有一个字节保存了颜色，但是同时%ah 寄存器了保存了颜色信息，二者感觉有点重复。